// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

interface ISEN is IERC20{

    function mint(address _to, uint256 _amount) external;

}

contract SenecaSale is Ownable{

    ISEN public sen;
    IERC20 public payToken;
    uint256 public saleCap;
    uint256 public minPurchase;
    uint256 public maxPurchase;
    uint256 public ratePer1e18;
    uint256 public vestingPeriod;
    uint256 public vestingCliff;
    uint256 public upfrontBPpostCliff;
    uint256 public totalSold;
    uint256 public totalClaimed;
    uint256 public tgeTime;
    uint8 public saleStage; // 0: pre-sale, 1: sale, 2: post-sale

    mapping(address => uint256) public amountPurchased;
    mapping(address => uint256) public amountClaimed;

    constructor(
        address _payToken,
        uint256 _saleCap,
        uint256 _minPurchase,
        uint256 _maxPurchase,
        uint256 _ratePer1e18,
        uint256 _vestingPeriod,
        uint256 _vestingCliff,
        uint256 _upfrontBPpostCliff
    ){
        payToken = IERC20(_payToken);
        saleCap = _saleCap;
        minPurchase = _minPurchase;
        maxPurchase = _maxPurchase;
        ratePer1e18 = _ratePer1e18;
        vestingPeriod = _vestingPeriod;
        vestingCliff = _vestingCliff;
        upfrontBPpostCliff = _upfrontBPpostCliff;
    }

    function invest(address to, uint256 amountToBuy) public{
        require(saleStage == 1, "SenecaSale : Sale not active");
        require(amountToBuy + amountPurchased[to] >= minPurchase, "SenecaSale : Amount too small");
        require(amountToBuy + amountPurchased[to] <= maxPurchase, "SenecaSale : Amount too large");
        require(totalSold + amountToBuy <= saleCap, "SenecaSale : Sale cap reached");
        uint256 payAmount = amountToBuy * ratePer1e18 / 1e18;
        payToken.transferFrom(msg.sender, owner(), payAmount);
        amountPurchased[to] += amountToBuy;
        totalSold += amountToBuy;
    }

    function claim() external{
        require(saleStage == 2, "SenecaSale : Sale not ended");
        require(tgeTime > 0, "SenecaSale : TGE time not set");
        require(amountPurchased[msg.sender] > 0, "SenecaSale : No purchase found");
        uint256 amountToClaim = _amountVested(msg.sender) - amountClaimed[msg.sender];
        sen.mint(msg.sender, amountToClaim);
        amountClaimed[msg.sender] += amountToClaim;
    }

    function amountVested(address investor) external view returns(uint256){
        return _amountVested(investor);
    }

    function _amountVested(address investor) internal view returns(uint256){
        if(block.timestamp < tgeTime + vestingCliff){
            return 0;
        }
        if(block.timestamp >= tgeTime + vestingPeriod){
            return amountPurchased[investor];
        }
        return ((amountPurchased[investor] * (block.timestamp - tgeTime - vestingCliff) *(10000 - upfrontBPpostCliff) ) / vestingPeriod / 10000) + (amountPurchased[investor] * upfrontBPpostCliff / 10000);
    }

    function setTgeTime(uint256 _tgeTime) external onlyOwner{
        require(saleStage == 2, "Sale already not over");
        require(tgeTime == 0, "TGE time already set");
        tgeTime = _tgeTime;
    }

    function setSaleStage(uint8 _saleStage) external onlyOwner{
        require(_saleStage >= saleStage, "Cannot go back in sale stage");
        require(_saleStage <= 2, "Invalid sale stage");
        saleStage = _saleStage;
    }

    function setMinPurchase(uint256 _minPurchase) external onlyOwner{
        minPurchase = _minPurchase;
    }

    function setMaxPurchase(uint256 _maxPurchase) external onlyOwner{
        maxPurchase = _maxPurchase;
    }

    function setSen(address _sen) external onlyOwner{
        sen = ISEN(_sen);
    }

    function emergencyWithdraw(address token, uint256 amount) external onlyOwner{
        IERC20(token).transfer(msg.sender, amount);
    }

}